# Resultant of two force vectors

• Vector 1 is 80 degrees counterclockwise from the X axis and 150 Newtons in magnitude
• Vector 2 is 15 degrees counterclockwise from the X axis and 100 Newtons in magnitude

What is the resultant vector?

Surely there is already a way that Mathematica 11.3 can compute this without me having to apply the law of sines and law of cosines.

• Welcome to Mma.SE. Start by taking the tour now and learning about asking and what's on-topic. Always edit if improvable, show due diligence, give brief context, include minimal working example of code and data in formatted form. By doing all this you help us to help you and likely you will inspire great answers. First share what have you tried already. – rhermans Aug 15 '18 at 15:08
• Thanks for your answer. I believe the question speaks for itself. Two vectors, how to get the resultant without having to manually use law of sines and cosines. In other words isn't there already a mathematica function to do this? – woops Aug 15 '18 at 15:46
• Look up AngleVector – Carl Woll Aug 15 '18 at 15:47
• F1:=AngleVector[Quantity[100,"Newtons"][15,"AngularDegrees"]] F2:=AngleVector[Quantity[150,"Newtons"][80,"AngularDegrees"]] Is this right? Then what? – woops Aug 15 '18 at 15:52

Define the vectors (the units are irrelevant for the angle you want to obtain in the end):

v1 = 100 AngleVector[15 Degree];
v2 = 150 AngleVector[80 Degree];


vec = v1 + v2 // FullSimplify


{25 (Sqrt[2] + Sqrt[6] + 6 Sin[10 °]), 25 (-Sqrt[2] + Sqrt[6] + 6 Cos[10 °])}

or

N[vec]


{122.64, 173.603}

The length is

Norm[vec] // FullSimplify


25 Sqrt[(-Sqrt[2] + Sqrt[6] + 6 Cos[10 °])^2 + (Sqrt[2] + Sqrt[6] + 6 Sin[10 °])^2]

N@Norm[vec]


212.552

You won't get a much simpler result because 10° does not result in a simple radical so that

angle = VectorAngle[vec, {1, 0}] // FullSimplify


ArcCos[(Sqrt[2] + Sqrt[6] + 6 Sin[10 °])/( 2 Sqrt[13 + 3 Sqrt[2] (-1 + Sqrt[3]) Cos[10 °] + 3 (Sqrt[2] + Sqrt[6]) Sin[10 °]])]

can be expressed only approximately

N@angle


0.955763

which is in radians. It can be easily converted to degrees:

UnitConvert[Quantity[N@angle, "Radians"], "Degrees"]


Quantity[54.7612, "AngularDegrees"]

or simply

angle 180./Pi


54.7612

• One could also use v1 = AngleVector[{100, 15 Degree}] etc. – Carl Woll Aug 15 '18 at 20:26
• corey979, Thank you for taking the time to show me how to work this in Mathematica. I would also like to say that your answer is indeed correct. Carl Woll, your AngleVector Form is noted too. Thanks. – woops Aug 16 '18 at 0:54